```
-- Rectangledefs.mesa  Edited by Sandman on March 28, 1978  5:01 PM

DIRECTORY
  SegmentDefs: FROM "segmentdefs";

RectangleDefs: DEFINITIONS =
  BEGIN

-- Magic memory locations and assoc. constants

  mousebuttonsup: CARDINAL = 7;

-- Constants for Display stuff

  leftmargin: CARDINAL = 10;
  blanklines: CARDINAL = 6; -- # of blank lines at the top
  maxwordsperline: CARDINAL = 38;
  maxbitsperline: CARDINAL = maxwordsperline*16;
  minwidth: CARDINAL = 32;
  minheight: CARDINAL = 32;

-- some TYPE's and null POINTERS

  restype: TYPE = {high, low};
  backgtype: TYPE = {white, black};
  xCoord: TYPE = [0..606];
  yCoord: TYPE = [0..808];

-- Display Hardware Stuff

  DCBchainHead: DCBptr = LOOPHOLE[420B];
  DCBnil: DCBptr = LOOPHOLE[0];
  DCBptr: TYPE = POINTER TO DCB;
  DCB: TYPE = MACHINE DEPENDENT RECORD [
    next: DCBptr,
    resolution: restype,
    background: backgtype,
    indenting: [0..77B],        -- in units of 16 bits
    width: [0..377B],   -- likewise; must be even
    bitmap: BMptr,              -- must be even
    height: CARDINAL];  -- in double scan lines

-- Font Stuff

  FHptr: TYPE = POINTER TO FontHeader;
  Fptr: TYPE = POINTER TO FONT;
  FCDptr: TYPE = POINTER TO FCD;
  FAptr: TYPE = POINTER TO fontarray;
  fontarray: TYPE = ARRAY [0..255] OF FCDptr;

  FONT: TYPE = MACHINE DEPENDENT RECORD [
    FHeader: FontHeader,
    FCDptrs: fontarray,         -- array of self-relative pointers to
      -- FCD's.  Indexed by char value.
      -- font pointer points hear!
    ExtFCDptrs: fontarray];     --  array of self-relative pointers to
      -- FCD's for extentions.  As large an
      -- array as needed.

  FontHeader: TYPE = MACHINE DEPENDENT RECORD [
    MaxHeight: CARDINAL,        -- height of tallest char in font (scan lines)
    VariableWidth: [0..1],      -- IF TRUE, proportionally spaced font
    blank:       [0..177B],     -- not used
    MaxWidth: [0..377B]];       -- width of widest char in font (raster units).

  FCD: TYPE = MACHINE DEPENDENT RECORD [
    widthORext: [0..77777B],    -- width or extention index
    HasNoExtension: BOOLEAN,    -- TRUE=> no ext.;prevfield=width
    height: [0..377B],          -- # scan lines to skip for char
    displacement: [0..377B]];   -- displacement back to char bitmap

-- Bit Map Stuff

  BitmapError: SIGNAL [bitmap: BMHandle, error: BitmapErrorCode];
  BitmapErrorCode: TYPE = {BitmapOperation};
  BITMAP: TYPE = WORD; -- array [0..15] of word
```

```
  BMptr: TYPE = POINTER TO BITMAP;
  BMHandle: TYPE = POINTER TO BitmapObject;

  BitmapObject: TYPE = RECORD [ -- all about a Bitmap
    link: BMHandle,     -- # NIL iff being displayed
    rectangles: Rptr, -- list of display streams for this map
    dcb: DCBptr,
    addr: BMptr,        -- it's address
    words: CARDINAL,    -- size of map (in words)
    wordsperline: [0..maxwordsperline],
    x0: xCoord,         -- x,y of upper left corner
    y0: yCoord,
    width: xCoord,
    height: yCoord,
    indenting: [0..77B],        -- in units of 16 bits
    resolution: restype,
    background: backgtype];

-- stuff for Bitmap Rectangles

  RectangleError: SIGNAL [rectangle:Rptr, error:RectangleErrorCode];
  RectangleErrorCode: TYPE = {RightOverflow, BottomOverflow, NotVisible};
  Rptr: TYPE = POINTER TO Rectangle;
  Rectangle: TYPE = RECORD [
    link: Rptr,
    visible: BOOLEAN,
    options: ROptions,
    bitmap: BMHandle,
    x0, width, cw: xCoord,
    y0, height, ch: yCoord];


  -- Rectangle options field definitions

    ROptions: TYPE = RECORD [   -- Rectangle options
      NoteInvisible: BOOLEAN,   -- SIGNAL if rectangle off bitmap
      NoteOverflow: BOOLEAN];   -- SIGNAL if storing outside

  GrayArray: TYPE = ARRAY [0..3] OF WORD;
  GrayPtr: TYPE = POINTER TO GrayArray;

-- Procedures Implementing RECTANGLES

  -- Bitmap Rectangle Routines

  CreateRectangle: PROCEDURE
    [bitmap: BMHandle, x0, width: xCoord, y0, height: yCoord]
    RETURNS[Rptr];
  DestroyRectangle: PROCEDURE[rectangle: Rptr];
  MoveRectangle: PROCEDURE [rectangle: Rptr, x: xCoord, y: yCoord];
  GrowRectangle: PROCEDURE [rectangle: Rptr, width: xCoord, height: yCoord];
  ClearBoxInRectangle: PROCEDURE
    [rectangle: Rptr, x0, width: xCoord, y0, height: yCoord, gray: GrayPtr];
  DrawBoxInRectangle: PROCEDURE
    [rectangle: Rptr, x0, width: xCoord, y0, height: yCoord];
  ScrollBoxInRectangle: PROCEDURE
    [rectangle: Rptr, x0, width: xCoord, y0, height: yCoord, incr: INTEGER];
  InvertBoxInRectangle: PROCEDURE
    [rectangle: Rptr, x0, width: xCoord, y0, height: yCoord];
  IsRectangleVisible: PROCEDURE [rectangle: Rptr] RETURNS [BOOLEAN];
  WriteRectangleChar: PROCEDURE
    [rectangle: Rptr, x: xCoord, y: yCoord, char: CHARACTER, pfont: FAptr]
    RETURNS[xCoord, yCoord];
  WriteRectangleString: PROCEDURE
    [rectangle: Rptr, x: xCoord, y: yCoord, str: STRING, pfont: FAptr]
    RETURNS[xCoord, yCoord];
  SaveRectangle:PROCEDURE [rectangle: Rptr] RETURNS [POINTER];
  RestoreRectangle:PROCEDURE [rectangle: Rptr, SegPtr: POINTER];

  -- Bitmap Routines

  GetDefaultBitmap: PROCEDURE RETURNS [BMHandle];
  CreateBitmap: PROCEDURE [pagesformap, wordsperline: CARDINAL]
    RETURNS [BMHandle];
  DestroyBitmap: PROCEDURE [mapdata: BMHandle] RETURNS [POINTER];
  UpdateBitmap: PROCEDURE [mapdata: BMHandle] RETURNS [DCBptr];
```

```
  ReallocateBitmap: PROCEDURE
    [mapdata: BMHandle, pagesformap, wordsperline: CARDINAL];
  DisplayBitmap: PROCEDURE [mapdata: BMHandle];
  UnDisplayBitmap: PROCEDURE [mapdata: BMHandle];
  CursorToMapCoords: PROCEDURE [mapdata: BMHandle, x: xCoord, y: yCoord]
    RETURNS [xCoord, yCoord];
  RectangleToMapCoords: PROCEDURE [rectangle: Rptr, x: xCoord, y: yCoord]
    RETURNS [xCoord, yCoord];
  CursorToRecCoords: PROCEDURE [rectangle: Rptr, x: xCoord, y: yCoord]
    RETURNS [xCoord, yCoord];


  -- Display Stuff

  DisplayOff: PROCEDURE [backgtype];
  DisplayOn: PROCEDURE;


  -- Font Stuff

  FileSegmentHandle: TYPE = SegmentDefs.FileSegmentHandle;

  ComputeCharWidth: PROCEDURE [char: CHARACTER, font: POINTER]
    RETURNS [CARDINAL];
  GetDefaultFont: PROCEDURE  RETURNS [FAptr, CARDINAL];
  GetFont: PROCEDURE [filename: STRING] RETURNS [FileSegmentHandle];
  LoadFont: PROCEDURE [segment: FileSegmentHandle] RETURNS [p: Fptr];

  RectanglesB: PROGRAM [pagesformap, mapwordsperline: CARDINAL];

  END.
```